
Soluţii

A. Execuţia programelor CLIPS. 

Să se deseneze graful reţelei RETE pe care o construieşte compilatorul CLIPS la compilarea regulilor următoare: 

(defrule R1 

  (patta ?x ?x aba ?y&:(integerp ?y))

  (patta ?u ?u abc ?x)

  (pattb ?x ?z aba ?u)

  =>

  …

)

(defrule R2 

  (patta ?u ?u aba)

  (pattb ?u ?y aba)

  =>

  …

)



Se acordă          puncte pentru desenarea corectă a reţelei comparaţiilor intra-şablon, respectiv     puncte pentru desenarea reţelei comparaţiilor inter-şabloane. 

B. Conceperea programelor CLIPS

În jocul „centrate” partenerul trebuie să ghicească, dintr-un număr cît mai mic de tentative, o secvenţă de 4 cifre pe care v-aţi ales-o. La fiecare pas el vă propune o secvenţă de 4 cifre, iar dv. trebuie să îi răspundeţi: 

· cîte cifre care sînt pe poziţiile lor a ghicit;

· cîte cifre care nu sînt pe poziţiile lor a ghicit. 

Scrieţi un set de fapte şi reguli CLIPS care să răspundă în locul dv. la o propunere a partenerului de joc.

Exemplu: 

Numărul meu: 2750

Numărul propus de partener: 3720

Răspunsul programului: Centrate: 2; Necentrate: 1.

; Rularea se face in doua faze: 

; - (faza cauta) – in care se adauga in baza atitea fapte de tipul 

;   (centrat <symbol-gen>) cite cifre centrate se recunosc intre cele doua

;   numere si atitea fapte de tipul (oricare <symbol-gen>) cite cifre

;   identice se recunosc intre cele doua numere, indiferent de pozitiile 

;   lor;
; - (faza numara) – in care se numara faptele (centrat …) si faptele 

;   (oricare …).

; In final se tipareste numarul faptelor centrate si diferenta dintre cele 

; oricare si centrate, ceea ce da numarul celor necentrate.  

; Faptul (potriviri <c> <o>) memoreaza in pozitia <c> numarul faptelor 

; (centrate …) si in pozitia <o> - numarul faptelor (oricare …).

(deffacts numere

   (numarul-meu 2 7 5 0)

   (numarul-tau 3 7 2 0)

   (faza cauta)

   (potriviri 0 0)

)

(defrule centrat

   (faza cauta)

   (numarul-meu $?beg1 ?x $?end1)

   (numarul-tau $?beg2&:(= (length $?beg1)(length $?beg2)) ?x $?end2)

   =>

   (assert (centrat =(gensym)))

)

(defrule oricare

   (faza cauta)

   (numarul-meu $?beg1 ?x $?end1)

   (numarul-tau $?beg2 ?x $?end2)

   =>

   (assert (oricare =(gensym)))

)

(defrule tranzitie

   ?f <- (faza cauta)

   =>

   (retract ?f)

   (assert (faza numara))

)

(defrule numara-centrate

   (faza numara)

   ?c <- (centrat ?)

   ?p <- (potriviri ?x ?y)

   =>

   (retract ?c ?p)

   (assert (potriviri =(+ ?x 1) ?y))

)

(defrule numara-oricare

   (faza numara)

   ?n <- (oricare ?)

   ?p <- (potriviri ?x ?y)

   =>

   (retract ?n ?p)

   (assert (potriviri ?x =(+ ?y 1)))

)

(defrule final

   (declare (salience -10))

   (potriviri ?x ?y)

   =>

   (printout t "Centrate: " ?x "; Necentrate: " (- ?y ?x) crlf)

)

În notare se va aprecia recunoaşterea capcanei unei soluţii în care se tentează numărarea directă a cifrelor centrate şi, respectiv, necentrate şi actualizarea unor contoare corespunzătoare (de exemplu prin utilizarea declaraţiei de prioritate care să prefere regula de recunoaştere a cifrele centrate, ce utilizează o condiţie de recunoaştere mai strictă, înaintea regulei de recunoaştere a cifrelor necentrate – mai laxă), ca în exemplul următor:

(deffacts numere

   (numarul-meu 2 7 5 0)

   (numarul-tau 3 7 2 0)

   (centrate 0)

   (necentrate 0)

)

(defrule centrate

   (declare (salience 10))

   (numarul-meu $?beg1 ?x $?end1)

   (numarul-tau $?beg2&:(= (length $?beg1)(length $?beg2)) ?x $?end2)

   ?idxc <- (centrate ?c)

   =>

   (retract ?idxc)

   (assert (centrate =(+ ?c 1)))

)

(defrule necentrate

   (numarul-meu $?beg1 ?x $?end1)

   (numarul-tau $?beg2 ?x $?end2)

   ?idxnc <- (necentrate ?nc)

   =>

   (retract ?idxnc)

   (assert (necentrate =(+ ?nc 1)))

)

(defrule final

   (declare (salience -10))

   (centrate ?c)

   (necentrate ?nc)

   =>

   (printout t "Centrate: " ?c "; Necentrate: " ?nc crlf)

)

Acest program ciclează pe regula centrate din motiv că faptul (centrate ...) este actualizat la fiecare aplicare. 

nod rădăcină





câmpul 2 stânga = câmpul 2 dreapta








câmpul 4 = aba





reţeaua comparaţiilor inter-şabloane





reţeaua comparaţiilor intra-şablon





câmpul 2 stânga = câmpul 5 dreapta











câmpul 1 = pattb





cîmpul 3 = cîmpul 2





R1 activată





cîmpul 1 = patta





câmpul 3 stânga = câmpul 5 dreapta


şi


câmpul 5 stânga = câmpul 2 dreapta








câmpul 4 = abc





R2 activată





(integerp cîmpul 5)





cîmpul 4 = aba





2





2





5








